
/*===================================================================================

==============* 

| Covance Study Number    : 000000106343                                                           | 

| Program Name            : t_anl_qsu.sas                                                          | 

| Purpose                 : Analysis of QSU                                

| 

| Input Data              : ADQSSU                                                             |                                                   

| Output Data             : T_15_02_04_53_01, T_15_02_04_53_02                                     |                                                                                                                                                                 

| Macros Called           : m_printto, m_logchk                                                    | 

| Originally Performed by : kpothuri                                                               | 

| Date                    : 11MAY2015                                                              |                                                                 

|                                                                                                  | 

|====================================================================================

==============| 

| Modification History                                                                             | 

|--------------------------------------------------------------------------------------------------| 

| Modified by          :                                                                           | 

| Modification Date    :                                                                           | 

| Modification Description  :                                                                      | 

+====================================================================================

==============*/ 

options notes source source2 nofullstimer validvarname=upcase missing=' ' NOQUOTELENMAX; 

ods _all_ close; 

ods listing; 

 

%m_printto(route=YES); 

 



*=====================================================================; 

* START OF PROGRAM CODE                                               ; 

*=====================================================================; 

 

proc datasets lib=work nolist memtype=data kill; quit; 

 

%let TFL_Part=%scan(&_SASPROGRAMFILE,-3,%str(/)); 

 

/* Standard - leave this */ 

data _null_; 

     tmp="&TFL_Part"; 

  if tmp not in ("dev" "qc") then call symput("TFL_Part", "prod"); 

  call symput('TFLpath', compress("&_SASPROGRAMFILE","'")); 

  call symput('TFLprg',reverse(scan(strip(reverse(compress("&_SASPROGRAMFILE","'"))),1,"/"))); 

run; 

 

 

*QSU; 

data QSU; 

 set adam.ADQSSU;  

 where DTYPE ne "LOCF" and ANL01FL="Y" and 

  avisit in ("Day 1", "Day 2", "Day 3", "Day 4", "Day 5", "Day 30", "Day 60", "Day 90");  

run; 

 

*QSU counts; 



%macro count (wh=, pop=); 

proc sort data=QSU out=QSU_mx&pop (where=(BASE ne . and UCPDGR1 ne '' and &wh)); by trtp parcat2 

avisit usubjid; run; *for proc mixed datasets; 

proc freq data=QSU_mx&pop noprint; 

 table trtp*parcat2*parcat2n*avisit*avisitn/out=f_QSU (drop=percent); 

run; 

proc sort data=f_QSU; by parcat2 avisit; run; 

proc transpose data=f_QSU out=t_QSU&pop (drop=_:);  

 id trtp; 

 var count; 

 by parcat2 parcat2n avisit avisitn; 

run; 

%mend count; 

*PP set; 

%count (wh=%str(PPROT1FL="Y" and avisit in ("Day 1", "Day 2", "Day 3", "Day 4", "Day 5")), pop=1); 

%count (wh=%str(PPROT2FL="Y" and avisit = "Day 30"), pop=2); 

%count (wh=%str(PPROT3FL="Y" and avisit = "Day 60"), pop=3); 

%count (wh=%str(PPROT4FL="Y" and avisit = "Day 90"), pop=4); 

 

 

*FAS set; 

%count (wh=%str(FASFL="Y" and ablfl ne "Y"), pop=_f); 

 

*PP set; 

data count_pp (drop=SA THSM2_2 MCC rename=(SA_=SA THS_=THSM2_2 MCC_=MCC)); 

length SA_ THS_ MCC_ $18 _name_ $10; 



 set t_QSU1 t_QSU2 t_QSU3 t_QSU4; 

 SA_=put(SA, best.); 

 THS_=put(THSM2_2, best.); 

 MCC_=put(MCC, best.); 

 

 num=0.5; 

 _name_="n"; 

run; 

 

data mix; *all pp set days; 

 set QSU_mx1 QSU_mx2 QSU_mx3 QSU_mx4; 

run; 

 

*FAS set; 

data count_f (drop=SA THSM2_2 MCC rename=(SA_=SA THS_=THSM2_2 MCC_=MCC)); 

length SA_ THS_ MCC_ $18 _name_ $10; 

 set t_QSU_f; 

 SA_=put(SA, best.); 

 THS_=put(THSM2_2, best.); 

 MCC_=put(MCC, best.); 

 

 num=0.5; 

 _name_="n"; 

run; 

 



 

*QSU stats - PP and FAS sets; 

%macro p (mx=, pop=); 

proc sort data=&mx; by parcat2 parcat2n avisit avisitn; run; 

proc mixed data=&mx;  

    by parcat2 parcat2n avisit avisitn; 

 Class trtp sex UCPDGR1; 

 Model aval = base sex UCPDGR1 trtp; 

 lsmeans trtp / pdiff =control('mCC') alpha=0.05 cl; 

    lsmeans trtp / pdiff =control('SA') alpha=0.05 cl; 

 ods output lsmeans=lsmeans (keep=parcat2 parcat2n avisit avisitn trtp lower upper estimate); 

*each arm; 

    ods output diffs=diffs (where=(trtp="THSm2.2") keep=parcat2 parcat2n avisit avisitn trtp _trtp lower 

upper estimate); *differences; 

run; 

 

*lsmean and C.I. for differences; 

data diffs_; 

length labnum $10; 

 set diffs; 

 format lower upper estimate; 

 if mod(_n_,2)=0 then labnum="SA"; 

  else labnum="MCC"; 

run; 

data LSM_CL; 

 set diffs_; 



  

 CI=compress(put(floor(100*lower)/100,12.2))||', '||compress(put(ceil(100*upper)/100,12.2)); 

 lsmean_=put(round(estimate,0.01),12.2); 

run; 

proc transpose data=LSM_CL out=t_LSM_CL (rename=(SA=ths_sa_diff mcc=ths_mcc_diff)); 

 id labnum; 

 var lsmean_ CI; 

 by parcat2 parcat2n avisit avisitn; 

run; 

 

*figure data; 

data fig&pop; 

length difftyp $25; 

 set LSM_CL; 

 

 if labnum="SA" then difftyp="THSm2.2vs.SA"; 

 if labnum="MCC" then difftyp="THSm2.2vs.mCC"; 

 keep parcat2 parcat2n avisit avisitn lower upper estimate difftyp; 

run; 

 

*lsmean, C.I.; 

proc sort data=lsmeans out=lsmeans_ nodupkey; by _all_; run; 

data lsmeans_; 

 set lsmeans_; 

 format lower upper estimate; 



run; 

data stat;  

 set lsmeans_; 

 

 lsmean_=put(round(estimate,0.01),12.2); 

 

 CI=compress(put(floor(100*Lower)/100,8.2))||', '||compress(put(ceil(100*Upper)/100,8.2));  

run; 

proc transpose data=stat out=t_stat; 

 id trtp; 

 var lsmean_ CI; 

 by parcat2 parcat2n avisit avisitn; 

run; 

 

*put stats together; 

proc sort data=t_LSM_CL; by parcat2 parcat2n avisit avisitn _name_; run; 

proc sort data=t_stat; by parcat2 parcat2n avisit avisitn _name_; run; 

data p; 

 merge t_LSM_CL t_stat; 

 by parcat2 parcat2n avisit avisitn _name_; 

 if _name_="LSMEAN_" then do; _name_="LS Mean"; num=1; end; 

 if _name_="CI" then do; _name_="95% CI"; num=2; end; 

run; 

proc sort data=p; by parcat2n avisitn num; run; 

 



data form; 

 set count&pop p; 

run; 

proc sort data=form; by parcat2n avisitn num; run; 

data form_1; 

 set form; 

 by parcat2n avisitn num; 

 if not first.avisitn then avisit=""; 

run; 

 

data dummy; 

length parcat2 $200 avisit $40; 

 num=0.2; parcat2="Factor 1 - Reward"; parcat2n=1; AVISIT="Factor 1 - Reward"; AVISITN=101; 

output; 

 num=0.2; parcat2="Factor 2 - Relief"; parcat2n=2; AVISIT="Factor 2 - Relief"; AVISITN=101; 

output; 

 num=0.2; parcat2="Total Score"; parcat2n=3; AVISIT="Total Score"; AVISITN=101; output; 

 num=0.2; parcat2="Factor 1 - Reward"; parcat2n=1; AVISIT="Factor 1 - Reward (cont...)"; 

AVISITN=105; output; 

 num=0.2; parcat2="Factor 2 - Relief"; parcat2n=2; AVISIT="Factor 2 - Relief (cont...)"; 

AVISITN=105; output; 

 num=0.2; parcat2="Total Score"; parcat2n=3; AVISIT="Total Score (cont...)"; AVISITN=105; 

output; 

run; 

 

data comb; 

 set dummy form_1; 



run; 

proc sort data=comb; by parcat2n avisitn num; run; 

 

data final&pop; 

 set comb; 

 if parcat2n=1 and avisitn in (101, 102, 103, 104) then pageord=1; 

 if parcat2n=1 and avisitn in (105, 130, 160, 190) then pageord=2; 

 if parcat2n=2 and avisitn in (101, 102, 103, 104) then pageord=3; 

 if parcat2n=2 and avisitn in (105, 130, 160, 190) then pageord=4; 

 if parcat2n=3 and avisitn in (101, 102, 103, 104) then pageord=5; 

 if parcat2n=3 and avisitn in (105, 130, 160, 190) then pageord=6; 

run; 

%mend p; 

%p (mx=mix, pop=_pp); 

%p (mx=QSU_mx_f, pop=_f); 

 

%let l_name = %str(L_15_04_04_53_01); 

%let t_title_l = %nrbquote(Listing 15.4.4.53.1 Analysis of QSU-brief Factors and Total Scores - PP Set); 

 

ods rtf  

file="/cvn/projects/prj/data/000000106343/TFL/dev/Tables/&l_name..rtf"; 

title "&t_title_l"; 

options orientation=landscape; 

%p (mx=mix, pop=_pp); 

ods rtf close; 



 

 

%let l_name2 = %str(L_15_04_04_53_02); 

%let t_title_2 = %nrbquote(Listing 15.4.4.53.2 Analysis of QSU-brief Factors and Total Scores - FAS); 

 

ods rtf  

file="/cvn/projects/prj/data/000000106343/TFL/dev/Tables/&l_name2..rtf"; 

title "&t_title_2"; 

options orientation=landscape; 

%p (mx=QSU_mx_f, pop=_f); 

ods rtf close; 

 

 

*Figure datasets; 

data tflds.T_15_02_04_53_01_F;  

 set fig_pp; 

run; 

 

data tflds.T_15_02_04_53_02_F;  

 set fig_f; 

run; 

 

 

%macro anlout (din=, tfl=, tabname=); 

 



%let tflno=&tfl.; 

 

data tflds.&tflno; 

 set &din end=last; 

 by pageord; 

 if last then call symputx("page", pageord); 

run; 

%put &page; 

 

/* Standard - leave this */ 

options number nodate orientation=landscape missing=' ';   

ods escapechar='$'; 

%let linetop = \brdrt\brdrs\brdrw30;  * needs to be 1.5pt so calculated in twips (1/20 pt) ; 

%let linebot = \brdrb\brdrs\brdrw30; 

 

 

/* Standard - macro for paging */ 

%macro outrtf(blankn=130, halfblnk=N); 

 

%if &halfblnk=N %then %let halfblnk=; 

%else %if &halfblnk=Y %then %let halfblnk=\~; 

 

 

ods path stdlib.t106343 (read) ; 

ods results off; 



ods rtf toc_data file="/cvn/projects/prj/data/000000106343/TFL/&TFL_Part./Tables/&tflno..rtf" 

style=t106343 startpage=yes headery=1440 footery=1440 ; 

ods noproctitle; 

%do i=1 %to &page; 

 

 

title ; 

footnote; 

%let wd=0; 

ods proclabel = ' '; 

 

 

data comp; 

     set tflds.&tflno end=eof;   

  where pageord=&i; 

 

        /* Amend title as needed */ 

  _firtitl="&tabname";   

  _upcas=(length("Path: &TFLpath.")-

length(compress("Path:&TFLpath.",'ABCDEFGHIJKLMNOPQRSTUVWXYZ')))/2; 

        len=&blankn.-length("(Page &i of &page)");  

  if eof then do; 

     call symput('_FSRTITL', trim(left(_firtitl))); 

  call symput('_blankn', compress(put(len,best.))); 

     end; 

  drop _firtitl  _upcas len; 



run; 

 

ods listing close; 

 

* most set up in template others below; 

* title arial 12pt bold with 12pt paragraph space below; 

* all headers to be arial 11pt bold; 

* data arial 10pt; 

* headers to be central, text values left aligned and numeric centered around decimal point; 

 

/* Update with your variables as needed */ 

proc report data = comp headline headskip nowd split = '$' %if &i=1 %then %do; contents=' ' %end; 

%else %do; contents='' %end;;; 

 column pageord avisitn avisit _name_ THSM2_2 mcc sa THS_MCC_DIFF THS_SA_DIFF;  

  define pageord  / order order = internal noprint; 

  define avisitn  / order order = internal noprint; 

  define avisit    /"Variable" display style={just=left cellwidth=3.0cm} 

style(header)={just=left}; 

        define _name_   /"Statistic" display style={just=left cellwidth=1.2cm} style(header)={just=left}; 

  define THSM2_2           /"THSm2.2" display style={just=c cellwidth=1.2cm} 

style(header)={just=center} ; 

  define mcc           /"mCC" display style={just=c cellwidth=1.2cm} 

style(header)={just=center}; 

  define sa           /"SA"  display style={JUST=c cellwidth=1.2cm} 

style(header)={just=center};      

  define THS_MCC_DIFF      /"THSm2.2 - mCC$ Difference" display style={just=c 

cellwidth=1.4cm} style(header)={just=center}; 



  define THS_SA_DIFF        /"THSm2.2 - SA$ Difference" display style={just=c 

cellwidth=1.4cm} style(header)={just=center}; 

 

 break after pageord / page; 

 

 compute after avisitn; 

  line " "; 

 endcomp; 

 

 compute before pageord / style={protectspecialchars=off};; 

     line "&linetop"; 

  endcomp; 

 

 compute before _page_ / style={just=left protectspecialchars=off};  

     line "\b\fs24\sa24&_FSRTITL." ;  * \b = bold, \fs24 is font size 12pt, \sa24 is space after 12pt; 

   line "&linebot"; 

  endcomp; 

 

  compute after _page_/ style={just=left protectspecialchars=off pretext="&linetop."};  

   line 'Note: Adjusted least squares (LS) means and confidence intervals (CIs) from an 

ANCOVA model conducted with baseline value, study arm, sex and mCC consumption reported at 

screening as fixed effect factors.'; 

  line 'Note: mCC = Menthol conventional cigarettes; SA = Smoking abstinence; THSm2.2 = 

Tobacco Heating System 2.2 Menthol.';   

  line '';   

  line 'Appendix 15.3.6.14'; 



  line "Study ID: ZRHM-REXA-08-US     Program: &TFLprg    Status: &status" 

&_blankn.*"\~\~" "&sysdate" &_blankn.*"\~\~" "(Page &i of &page)";  

  endcomp; 

run; 

%end; 

ods rtf close; 

ods results on; 

ods path sashelp.tmplmst (read); 

 

%mend ; 

 

%outrtf(blankn=36, halfblnk=N); 

 

ods listing; 

 

%mend anlout; 

%anlout (din=final_pp, tfl=%str(T_15_02_04_53_01), tabname=%str(Table 15.2.4.53.1 Analysis of 

QSU-brief Factors and Total Scores - PP Set)); 

%anlout (din=final_f, tfl=%str(T_15_02_04_53_02), tabname=%str(Table 15.2.4.53.2 Analysis of 

QSU-brief Factors and Total Scores - FAS)); 

 

*=====================================================================; 

* END OF PROGRAM CODE                                                 ; 

*=====================================================================; 

 

%m_logchk; 

 



 

 

 

 

 

 

 

 


