
/*===================================================================================

==============* 

| Covance Study Number        : 000000106343                                                           

| 

| Program Name                     : t_cigbrand.sas                                                               

| 

| Purpose                         : Create table of 

summary of   current cigarette brand at screening - safety population                                           | 

| Input Data                      : ADAM.ADFA 

ADAM.ADSL                                                        | 

| Output Data                    : T_15_02_01_05                                                          

|         | 

| Macros Called                  :%m_printto , %m_logchk                                                                     

| 

| Originally Performed by      : Upender                                                             

| 

| Date                                 : 20Apr2015                                                              

| 

|====================================================================================

==============| 

| Modification History : Original Version                                                          | 

|--------------------------------------------------------------------------------------------------| 

| Modified by          :                                                                                   

          | 

| Modification Date    :                                                                           | 

| Modification Reason  :                                                                           | 

+====================================================================================

==============*/ 

proc datasets lib=work nolist memtype=data kill; quit; 

options notes source source2 nofullstimer validvarname=upcase missing=' '; 



ods _all_ close; 

ods listing; 

 

%m_printto(route=YES); 

*=====================================================================; 

* START OF PROGRAM CODE                                               ; 

*=====================================================================; 

 

%let tflno=T_15_02_01_05; 

 

%let TFL_Part=%scan(&_SASPROGRAMFILE,-3,%str(/)); 

 

data _null_; 

     tmp="&TFL_Part"; 

  if tmp not in ("dev" "qc") then call symput("TFL_Part", "prod"); 

  call symput('TFLpath', compress("&_SASPROGRAMFILE","'")); 

   call symput('TFLprg',reverse(scan(strip(reverse(compress("&_SASPROGRAMFILE","'"))),1,"/"))); 

 

run; 

 

******************************; 

* read in data ; 

******************************; 

 

/*Bring in ADSL for column headers*/ 



proc sort data=adam.adsl(where = (safafl = 'Y' or safbfl='Y') ) out=adsl; 

 by usubjid; 

run; 

 

data adsl2a; 

 set adsl; 

 if index(trt01a,'Exposed') or missing(trt01a)then delete; 

  output; 

 trt01a='Overall Safety'; 

 trt01an=99; 

  output; 

  keep usubjid trt01a trt01an ; 

run; 

 

 

proc sql; 

 

select count(distinct usubjid) into: trt4 from adsl2a (where=(trt01an = 4 )); 

select count(distinct usubjid) into: trt5 from adsl2a (where=(trt01an = 5 )); 

select count(distinct usubjid) into: trt3 from adsl2a (where=(trt01an = 3 )); 

select count(distinct usubjid) into: trt96 from adsl2a (where=(trt01an = 96 )); 

select count(distinct usubjid) into: trt99 from adsl2a (where=(trt01an = 99 )); 

 

 

quit; 



 

%put tt4= &trt4. , tt5=&trt5. , tt3=&trt3. , tt96=&trt96. ;  

 

 

/* Current Cigarette Brand data */ 

proc sort data=adam.adfa  (where = ((safafl = 'Y' or safbfl='Y')   and EPOCH='SCREENING' and paramcd in 

('TYIELD')   )) out=adfa;    

 by trtan trta subjidn brand; 

run; 

 

 

data adfa_bigN; 

set adfa; 

output; 

trtan=99; 

trta='Overall Safety'; 

output; 

run; 

 

proc sql; 

 

select count(distinct usubjid) into: strt4 from adfa_bigN (where=(trtan = 4 )); 

select count(distinct usubjid) into: strt5 from adfa_bigN (where=(trtan = 5 )); 

select count(distinct usubjid) into: strt3 from adfa_bigN (where=(trtan = 3 )); 

select count(distinct usubjid) into: strt96 from adfa_bigN (where=(trtan = 96 )); 

select count(distinct usubjid) into: strt99 from adfa_bigN (where=(trtan = 99 )); 



 

 

quit; 

 

%put stt4= &strt4. , stt5=&strt5. , stt3=&strt3. , stt96=&strt96.  , stt99=&strt99. ;  

 

proc freq data=adfa_bigN noprint; 

tables trtan /out=smlN (drop=percent); 

run; 

 

proc transpose data=smlN out=smallfreq (drop=_label_)  prefix=trt; 

 var count; 

 id trtan; 

run; 

 

data smlfreq1 (Drop=trt3 trt4 trt5 trt96 trt99 _name_ rename=(mtrt3=trt3 mtrt4=trt4 mtrt5=trt5 

mtrt96=trt96 mtrt99=trt99)) ; 

set smallfreq; 

if trt3 ne . then mtrt3=&trt3. - trt3; 

if trt4 ne . then mtrt4=&trt4. - trt4; 

if trt5 ne . then mtrt5=&trt5. - trt5; 

if trt96 ne . then mtrt96=&trt96. - trt96; 

if trt99 ne . then mtrt99=&trt99. - trt99; 

array m6 (5) mtrt3 mtrt4 mtrt5 mtrt96 mtrt99; 

do i=1 to 5; 

if m6(i)=0 then m6(i)=. ; 



end; 

Brand_New='Missing'; 

ord=999.5; 

run; 

 

 

data adfa1; 

set adfa; 

keep trta trtan brand subjid; 

run; 

 

 

 

proc sort data=adfa1 out=adfa2  ; by trtan trta subjid brand; run; 

 

 

proc freq data=adfa2; 

   table brand / noprint out=brand(drop=percent); 

run; 

 

 

proc sort data=brand; by brand; run; 

proc sort data=adfa2; by brand; run; 

 

data adfa3; 



length brand_new $80.; 

merge adfa2 (in=a) brand (in=b); 

by brand;  

if count lt 4 then brand_new='OTHER'; 

else brand_new=brand; 

run; 

 

data adfa4;  

 set adfa3; 

 output; 

 trtan=99; 

 trta='Overall Safety'; 

 output; 

run; 

 

 

proc freq data=adfa4 (drop=brand where=(brand_new ne '')); 

tables brand_new*trta*trtan /out=brandfreq (drop=percent); 

run; 

 

proc transpose data=brandfreq out=bfreqs (drop=_:) prefix=trt;  

by brand_new; 

id trtan ; 

idlabel trta; 

var count; 



run; 

 

proc sort data=bfreqs (keep=brand_new trt99) out=order; by descending trt99 brand_new; run; 

 

data order1; 

set order; 

by descending trt99 brand_new; 

ord+1; 

run; 

 

proc sort data=bfreqs out=bfreqs1 ; by brand_new; run; 

proc sort data=order1 ; by brand_new; run; 

 

 

data final_a; 

merge bfreqs1 (in=a) order1 (in=b drop=trt99); 

by brand_new;  

run; 

 

data final_b (drop=brand_new rename=(brand_new1=brand_new)); 

set final_a; 

if brand_new='OTHER' then ord=999; 

brand_new1=propcase(brand_new); 

run; 

 



 

proc sort data=final_b; by ord; run; 

 

 

data final; 

set final_b smlfreq1; 

by ord; 

run; 

 

proc sort data=final (where=(ord ne 999.5)) ; by ord; run; 

 

data final1; 

set final; 

if ^missing(trt99) then do; 

OS=put(trt99, best.)||' ('||put((trt99/&trt99.)*100, 5.1)||')'; 

end; 

 

if ^missing(trt3) then do; 

SA=put(trt3, best.)||' ('||put((trt3/&trt3.)*100, 5.1)||')'; 

end; 

 

if ^missing(trt4) then do; 

THS=put(trt4, best.)||' ('||put((trt4/&trt4.)*100, 5.1)||')'; 

end; 

 



if ^missing(trt5) then do; 

MCC=put(trt5, best.)||' ('||put((trt5/&trt5.)*100, 5.1)||')'; 

end; 

 

if ^missing(trt96) then do; 

PT=put(trt96, best.)||' ('||put((trt96/&trt96.)*100, 5.1)||')'; 

end; 

run; 

 

data tflds.T_15_02_01_05 (drop= trt3 trt4 trt5 trt96 trt99); 

set final1; 

run; 

 

 

 

data paging; 

  set final1; 

 by ord; 

 

 flag=1; 

 

  if ln gt 15 then ln=1;    

  else ln+1; 

  if ln=1 then page+1; 

  call symput("page",compress(put(page,best.))); 



run; 

 

options number nodate orientation=landscape missing=' '; 

ods escapechar='$'; 

%let linetop = \brdrt\brdrs\brdrw30;  * needs to be 1.5pt so calculated in twips (1/20 pt) ; 

%let linebot = \brdrb\brdrs\brdrw30; 

 

%macro outrtf(blankn=, halfblnk=); 

 

%if &halfblnk=N %then %let halfblnk=; 

%else %if &halfblnk=Y %then %let halfblnk=\~; 

 

 

ods path stdlib.t106343 (read) ; 

ods rtf toc_data/* contents*/ 

file="/cvn/projects/prj/data/000000106343/TFL/&TFL_Part./Tables/&tflno..rtf" style=t106343 

startpage=yes headery=1440 footery=1440 ; 

ods noproctitle; 

%do i=1 %to &page; 

 

 

title ; 

footnote; 

%let wd=0; 

ods proclabel = ' '; 

 



data comp; 

     set paging end=eof; 

   where page=&i; 

      

     /* Amend title as needed */ 

  _firtitl="Table 15.2.1.5 Summary of Current Cigarette Brands at Screening - Safety Population"; 

  _upcas=(length(_firtitl)-length(compress(_firtitl,'ABCDEFGHIJKLMNOPQRSTUVWXYZ')))/2; 

     len=&blankn.-length("(Page &i of &page)"); 

  if eof then do; 

     call symput('_FSRTITL', trim(left(_firtitl))); 

  call symput('_blankn', compress(put(len,best.))); 

  call symput('N3', strip(put(&trt3., best.))); 

  call symput('N4', strip(put(&trt4., best.))); 

  call symput('N5', strip(put(&trt5., best.))); 

  call symput('N96', strip(put(&trt96., best.))); 

  call symput('N99', strip(put(&trt99., best.))); 

  end; 

  drop _firtitl  _upcas len; 

run; 

 

 

 

* most set up in template others below; 

* title arial 12pt bold with 12pt paragraph space below; 

* all headers to be arial 11pt bold; 



* data arial 10pt; 

* headers to be central, text values left aligned and numeric centered around decimal point; 

ods listing close; 

proc report data = comp headline headskip missing nowd split = '#' %if &i=1 %then %do; contents=' ' 

%end; %else %do; contents='' %end;;;; 

 column flag page brand_new THS MCC SA PT OS ;  

 

  define flag         / order order=internal noprint; 

 define page        / order order = internal noprint; 

  define brand_new        / display style={just=left cellwidth=3cm} "Brand"; 

 

 define THS           / "THSm2.2#(N=&N4.)#n (%)" display style={just=center cellwidth=1.2cm} 

style(header)={just=center} ;   

 define MCC          / "mCC#(N=&N5.)#n (%)" display style={just=center cellwidth=1.2cm} 

style(header)={just=center};   

 define SA           / "SA#(N=&N3.)#n (%)" display style={just=center cellwidth=1.2cm} 

style(header)={just=center} ;   

 define PT          / "Product Test#(N=&N96.)#n (%)" display style={just=center cellwidth=1.2cm} 

style(header)={just=center};   

 define OS          / "Overall Safety#(N=&N99.)#n (%)" display style={just=center cellwidth=1.2cm} 

style(header)={just=center} ;   

 

 break before flag / page %if &i=1 %then %do;  

 contents="&_fsrtitl" %end; %else %do; contents='' %end;; 

  

 break after page / page; 

 

 compute before page / style={just=left cellwidth=5cm protectspecialchars=off}; 



     line "&linetop"; 

  endcomp; 

 

  compute after page / style={just=left cellwidth=5cm protectspecialchars=off}; 

     line "&linebot" ; 

  endcomp; 

 

  compute before _page_ / style={just=left protectspecialchars=off};  

     line "\b\fs24\sa24&_FSRTITL." ;  * \b = bold, \fs24 is font size 12pt, \sa24 is space after 12pt; 

  line "&linebot"; 

  endcomp; 

 

  compute after _page_ / style={just=left protectspecialchars=off}; 

   line "Note: 'Product Test' refers to all subjects who tested the THS product but were not 

randomized."; 

   line 'Note: The Overall Safety refers to all subjects in the Safety Population.'; 

   line 'Note: mCC = Menthol Conventional cigarettes; SA = Smoking abstinence; THSm2.2 

= Tobacco Heating System 2.2 Menthol.'; 

  line 'Note: Percentages are based on the number of subjects indicated in the column 

header (N).';  

  line 'Note: Only brands used by at least 4 subejcts in the Safety Population are 

presented.';  

  line "";  

 

    line "Appendix 15.3.1.2"; 

 line "Study ID:ZRHM-REXA-08-US        Program: &TFLprg          Status: &status" &_blankn.*"\~\~" 

"&sysdate" &_blankn.*"\~\~" "(Page &i of &page)";  



  endcomp; 

run; 

%end; 

ods rtf close; 

ods results on; 

ods path sashelp.tmplmst (read); 

 

%mend ; 

 

%outrtf(blankn=30, halfblnk=N); 

 

%m_logchk; 

 

 

*=====================================================================; 

* END OF PROGRAM CODE                                                 ; 

*=====================================================================; 

 


