
/*=================================================================================* 

| Covance Study Number    : COV-  106331                                           | 

| Client Protocol ID      : ZRHM-REXA-07-JP                                        | 

| Program Name            : t_desc_nc.sas                                           | 

| Purpose                 : Table Descriptive statistics of plasma nicotine and cotinine       | 

| Input Data              : ADBX ADSL                                         | 

| Output Data             : T_15_02_04_19_01,19_02                                 | 

|                                                                                  | 

| Macros Called           :                                                        | 

|                                                                                  | 

| Originally Performed by : Seroan Zheng                                           |  

| Date/Time billed        : 12May2015                                              | 

|                                                                                  | 

+==================================================================================+ 

| Modification History                                                             | 

|                                                                                  | 

| Programmer              : Serona Zheng                                           | 

| Date                    : 25Aug2015                                              | 

| Reason for Change       : 1. Change day 5 08:00PM - 10:00 PM order after day T0 + 16h| 

|                           2. Change avisit length to avoid content cutting       | 

|                                                                                  | 

+=================================================================================*/ 

options noquotelenmax; 

***Create log file; 

proc printto  new 



log="/cvn/projects/prj/development/000000106343/dev/tables/log/t_desc_nc.log"  ; 

run; 

 

%macro t_desc_nc(t_name=,l_name=,dsin=,conf1=,by_var1=,t_title=,t_title_l=,append=,pop=); 

%let pgname=t_desc_nc.sas; 

%let table=&tflout.;  

%let tflno=&t_name; 

%let TFL_Part=%scan(&_SASPROGRAMFILE,-3,%str(/)); 

 

*** Standard - leave this ; 

 

data _null_; 

     tmp="&TFL_Part"; 

  if tmp not in ("dev" "qc") then call symput("TFL_Part", "prod"); 

  call symput('TFLpath', compress("&_SASPROGRAMFILE","'")); 

run; 

 

 

proc sql; 

%if &pop= pp %then %do; 

select count(distinct usubjid) into: N1THS from adam.adsl(where=(trt01an = 4 and pprot1fl = "Y")); 

select count(distinct usubjid) into: N1MCC from adam.adsl(where=(trt01an = 5 and pprot1fl = "Y")); 

select count(distinct usubjid) into: N1SAA from adam.adsl(where=(trt01an = 3 and pprot1fl = "Y")); 

 

select count(distinct usubjid) into: N2THS from adam.adsl(where=(trt01an = 4 and pprot2fl = "Y")); 



select count(distinct usubjid) into: N2MCC from adam.adsl(where=(trt01an = 5 and pprot2fl = "Y")); 

select count(distinct usubjid) into: N2SAA from adam.adsl(where=(trt01an = 3 and pprot2fl = "Y")); 

 

select count(distinct usubjid) into: N3THS from adam.adsl(where=(trt01an = 4 and pprot3fl = "Y")); 

select count(distinct usubjid) into: N3MCC from adam.adsl(where=(trt01an = 5 and pprot3fl = "Y")); 

select count(distinct usubjid) into: N3SAA from adam.adsl(where=(trt01an = 3 and pprot3fl = "Y")); 

 

select count(distinct usubjid) into: N4THS from adam.adsl(where=(trt01an = 4 and pprot4fl = "Y")); 

select count(distinct usubjid) into: N4MCC from adam.adsl(where=(trt01an = 5 and pprot4fl = "Y")); 

select count(distinct usubjid) into: N4SAA from adam.adsl(where=(trt01an = 3 and pprot4fl = "Y")); 

%end; 

%else %do; 

select count(distinct usubjid) into: N1THS from adam.adsl(where=(trt01an = 4 and fasfl = "Y")); 

select count(distinct usubjid) into: N1MCC from adam.adsl(where=(trt01an = 5 and fasfl = "Y")); 

select count(distinct usubjid) into: N1SAA from adam.adsl(where=(trt01an = 3 and fasfl = "Y")); 

 

select count(distinct usubjid) into: N2THS from adam.adsl(where=(trt01an = 4 and fasfl = "Y")); 

select count(distinct usubjid) into: N2MCC from adam.adsl(where=(trt01an = 5 and fasfl = "Y")); 

select count(distinct usubjid) into: N2SAA from adam.adsl(where=(trt01an = 3 and fasfl = "Y")); 

 

select count(distinct usubjid) into: N3THS from adam.adsl(where=(trt01an = 4 and fasfl = "Y")); 

select count(distinct usubjid) into: N3MCC from adam.adsl(where=(trt01an = 5 and fasfl = "Y")); 

select count(distinct usubjid) into: N3SAA from adam.adsl(where=(trt01an = 3 and fasfl = "Y")); 

 

select count(distinct usubjid) into: N4THS from adam.adsl(where=(trt01an = 4 and fasfl = "Y")); 



select count(distinct usubjid) into: N4MCC from adam.adsl(where=(trt01an = 5 and fasfl = "Y")); 

select count(distinct usubjid) into: N4SAA from adam.adsl(where=(trt01an = 3 and fasfl = "Y")); 

%end; 

 

quit; 

 

***Get raw data; 

data anl1; 

 set adam.&dsin; 

  &conf1 

 

***Calculate LOG; 

  if aval not in (.,0) then do; 

        logaval=log(aval); 

  if base not in (.,0) then logbase=log(base); 

  end; 

 

  if avisitn = 105 and atptn ne . then do; 

  avisitn_ = strip(put(avisitn,best.))||'.'||strip(put(atptn,best.)); 

  avisitn = input(avisitn_,best.); 

 end; 

run; 

proc sort data=anl1;by &by_var1;run; 

 

data base1; 



 set adam.adpc; 

  if aval not in (.,0) then do; 

        logaval=log(aval); 

  end; 

 if paramcd in ('NIC' 'COT') and anl01fl = "Y"; 

 %if &pop = pp %then %do; 

  if ablfl = "Y" and pprot1fl = "Y" then do; 

   avisitn = 10; 

   avisit = "Baseline"; 

   apuper = 1; 

   apuperc = "Period 1"; 

   output; 

  end; 

  if ablfl = "Y" and pprot2fl = "Y" then do; 

   avisitn = 10; 

   avisit = "Baseline"; 

   apuper = 2; 

   apuperc = "Period 2"; 

   output; 

  end; 

  if ablfl = "Y" and pprot3fl = "Y" then do; 

   avisitn = 10; 

   avisit = "Baseline"; 

   apuper = 3; 

   apuperc = "Period 3"; 



   output; 

  end; 

  if ablfl = "Y" and pprot4fl = "Y" then do; 

   avisitn = 10; 

   avisit = "Baseline"; 

   apuper = 4; 

   apuperc = "Period 4"; 

   output; 

  end; 

 %end; 

 %else %if &pop = fas %then %do; 

  if ablfl = "Y" and fasfl = "Y" then do; 

   avisitn = 10; 

   avisit = "Baseline"; 

   apuper = 1; 

   apuperc = "Period 1"; 

   output; 

  end; 

 %end; 

run; 

***Get decimal length;  

data temp; 

    set anl1; 

    declen=lengthn(scan(strip(put(aval, best.)),2,".")); 

run; 



 

proc sql; 

    create table dectemp 

    as select distinct paramn, max(declen) as declen 

    from temp 

    group by paramn; 

 

 select max(declen) into: last 

 from dectemp; 

quit; 

 

data anl2; 

 set anl1 base1; 

run; 

 

proc sort data=anl2; 

 by trtpn paramn param avalu avalu apuper apuperc avisitn avisit atptn atpt ; 

run; 

 

proc means data=anl2(where=(aval ne .)) noprint; 

 var aval; 

 by trtpn paramn param avalu apuper apuperc avisitn avisit atptn atpt ; 

 output out=aval n =n mean = mean std = std median = median min = min max = max q1 = q1 q3 

= q3 lclm = lclm uclm = uclm; 

run; 

proc sort data=aval;by paramn;run; 



data aval1; 

 merge aval(in=a) dectemp; 

 by paramn; 

 if a; 

 %do i=0 %to &last.; 

  if declen=%eval(&i) then do; 

  %let fmt= %sysevalf(12 + (&i +1 )*0.1); 

  %if &i = 0 %then %let fmt1 = 12.; 

  %else  

  %let fmt1= %sysevalf(12 + (&i)*0.1); 

  %let fmt2= %sysevalf(12 + (&i +2 )*0.1); 

 

 if lclm ne . then lclmx = floor(10**(declen+1)*lclm)/10**(declen+1); 

 if uclm ne . then uclmx = ceil(10**(declen+1)*uclm)/10**(declen+1); 

 length median1 Q2575 Minmax Meansd CIAM n1 $50.; 

 if median ne . then median1 = strip(put(round(median,1/10**(declen+1)), &fmt.)); 

 q2575 = strip(put(round(q1, 1/10**(declen+1)), &fmt.))||", "||strip(put(round(q3, 

1/10**(declen+1)), &fmt.)); 

 minmax = strip(put(round(min, 1/10**(declen)), &fmt1.))||", 

"||strip(put(round(max,1/10**(declen)), &fmt1.)); 

 if std ne . then meansd = strip(put(round(mean, 1/10**(declen+1)), &fmt.))||" 

("||strip(put(ceil(10**(declen+2)*std)/10**(declen+2), &fmt2.))||")"; 

 else meansd = strip(put(round(mean, 1/10**(declen+1)),&fmt.))||" (NA)";  

 if nmiss(lclm, uclm) = 0 then ciam = strip(put(lclmx, &fmt.))||", "||strip(put(uclmx, &fmt.)); 

 else if lclm = . and uclm ne . then ciam = "NA, "||strip(put(uclmx, &fmt.)); 

 else if lclm ne . and uclm = . then ciam = strip(put(lclmx, &fmt.))||", NA"; 

 else if lclm = . and uclm = . then ciam = "NA, NA"; 



 n1 = strip(put(n, best.)); 

 end; 

 %end; 

 %do i = 1 %to 4; 

 if apuper = &i then do; 

 if trtpn = 3 then Num = &&N&i.SAA; 

 if trtpn = 4 then Num = &&N&i.THS; 

 if trtpn = 5 then Num = &&N&i.MCC; 

 end; 

 %end; 

 missing = Num - n; 

 if missing ne 0 then miss = strip(put(missing, best.))||" 

("||strip(put(round((missing/Num)*100,0.1),10.1))||")"; 

 else miss = '0'; 

run; 

 

proc sort data=aval1; 

 by paramn param avalu apuper apuperc avisitn avisit atptn atpt ; 

run; 

 

 

proc transpose data=aval1 out=aval_t prefix= trt_; 

 by paramn param avalu apuper apuperc avisitn avisit atptn atpt ; 

 var n1 median1 q2575 minmax meansd ciam miss; 

 id trtpn; 

run; 



 

data aval_t; 

 length txt $200.; 

 set aval_t; 

 if upcase(_name_) = "N1" then do; 

 txtn = 1; 

 txt = "n"; 

 end; 

 if upcase(_name_) = "MISS" then do; 

 txtn = 2; 

 txt = "Missing, n (%)"; 

 end; 

 else if upcase(_name_) = "MEDIAN1" then do; 

 txtn = 7; 

 txt = "Median"; 

 end; 

 else if upcase(_name_) = "Q2575" then do; 

 txtn = 8; 

 txt = "Q25, Q75"; 

 end; 

 else if upcase(_name_) = "MINMAX" then do; 

 txtn = 9; 

 txt = "Min, Max"; 

 end; 

 else if upcase(_name_) = "MEANSD" then do; 



 txtn = 10; 

 txt = "Mean (SD)"; 

 end; 

 else if upcase(_name_) = "CIAM" then do; 

 txtn = 11; 

 txt = "95% CI of Mean"; 

 end; 

run; 

 

proc sort data=anl2; 

 by trtpn paramn param avalu apuper apuperc avisitn avisit atptn atpt ; 

run; 

 

proc means data=anl2 noprint; 

 by trtpn paramn param avalu apuper apuperc avisitn avisit atptn atpt ; 

 output out=aval_log mean = mean std = std lclm = lclm uclm = uclm; 

 var logaval; 

run; 

proc sort data=aval_log;by paramn;run; 

data aval_log1; 

 length geocv CIGM $50.; 

 merge aval_log(in=a) dectemp; 

 by paramn; 

 if a; 

 %do i=0 %to &last.; 



  if declen=%eval(&i) then do; 

   %let fmt= %sysevalf(12 + (&i +1 )*0.1); 

   %if &i = 0 %then %let fmt1 = 12.; 

   %else  

   %let fmt1= %sysevalf(12 + (&i)*0.1); 

   %let fmt2= %sysevalf(12 + (&i +2 )*0.1); 

   if lclm ne . then lclm1 = exp(lclm); 

   if uclm ne . then uclm1 = exp(uclm); 

   if lclm1 ne . then lclmx = floor(10**(declen+1)*lclm1)/10**(declen+1); 

   if uclm1 ne . then uclmx = ceil(10**(declen+1)*uclm1)/10**(declen+1); 

   if std ne . then cv_pct=100*sqrt(exp(std*std)-1) ; 

   if mean ne . then mean1 = exp(mean); 

   if cv_pct ne . and mean1 ne . then geocv = 

strip(put(round(mean1,1/10**(declen+1)), &fmt.))||" 

("||strip(put(ceil(10**(declen+2)*cv_pct)/10**(declen+2), &fmt2.))||")"; 

   else if mean ne . then geocv = strip(put(round(mean,1/10**(declen+1)), 

&fmt.))||" (NA)"; 

   if nmiss(lclmx, uclmx) = 0 then CIGM = strip(put(lclmx, &fmt.))||", 

"||strip(put(uclmx, &fmt.)); 

   else if lclm = . and uclmx ne . then CIGM = "NA, "||strip(put(uclmx, &fmt.)); 

   else if lclmx ne . and uclm = . then CIGM = strip(put(lclmx, &fmt.))||", NA"; 

   else if lclm = . and uclm = . then CIGM = "NA, NA"; 

  end; 

 %end; 

run; 

 

proc sort data=aval_log1 ; 



 by  paramn param avalu apuper apuperc avisitn avisit atptn atpt; 

run; 

 

proc transpose data=aval_log1 out=aval_log1_t prefix= trt_; 

 by  paramn param avalu apuper apuperc avisitn avisit atptn atpt ; 

 var geocv cigm; 

 id trtpn; 

run; 

 

data aval_log1_t; 

 length txt $200.; 

 set aval_log1_t; 

 if upcase(_name_) = "GEOCV" then do; 

 txtn = 5; 

 txt = "Geometric Mean (CV%)"; 

 end; 

 else if upcase(_name_) = "CIGM" then do; 

 txtn = 6; 

 txt = "95% CI of Geometric Mean"; 

 end; 

run; 

 

***Count BLOQ number; 

data n; 

 length txt $200.; 



 set anl2; 

 where bloqfl = 'Y'; 

 txtn = 3; 

 txt = "BLOQ, n(%)"; 

run;  

 

proc sort data=n out=n1 nodupkey dupout=dup; 

 by usubjid paramn avalu avisitn txtn atpt; 

run; 

 

proc freq data=n1 noprint; 

 tables 

paramn*param*avalu*apuper*apuperc*avisitn*avisit*atptn*atpt*txtn*txt*trtpn/out=n_freq; 

run; 

 

data timep_n; 

 set aval; 

 keep trtpn paramn param avalu apuper apuperc avisitn avisit atptn atpt n; 

 if avisit ne 'Baseline'; 

run; 

proc sort data=timep_n;by paramn param avalu apuper apuperc avisitn avisit atptn atpt trtpn;run; 

data n_freq; 

 set n_freq; 

 keep trtpn paramn param avalu apuper apuperc avisitn avisit atptn atpt count txtn txt; 

run; 

proc sort data=n_freq;by paramn param avalu apuper apuperc avisitn avisit atptn atpt trtpn;run; 



 

data n_freq; 

 length countx $50.; 

 merge n_freq(in=a) timep_n(in=b); 

 by paramn param avalu apuper apuperc avisitn avisit atptn atpt trtpn; 

 if a; 

 countx = strip(put(count,best.))||" ("||strip(put(round((count/n)* 100, 0.1), 15.1))||")"; 

run; 

 

proc sort data=n_freq; 

 by paramn param avalu apuper apuperc avisitn avisit atptn atpt txtn txt; 

run; 

 

proc transpose data=n_freq out=n_freq_t prefix=trt_; 

 by paramn param avalu apuper apuperc avisitn avisit atptn atpt txtn txt; 

 var countx; 

 id trtpn; 

run; 

 

data n_freq_t; 

 set n_freq_t; 

 if paramn ne .; 

 if trt_3 = '' then trt_3 = '0'; 

 if trt_4 = '' then trt_4 = '0'; 

 if trt_5 = '' then trt_5 = '0'; 



run; 

 

***combine all dataset together; 

data aval_f; 

 set aval_t n_freq_t aval_log1_t; 

 %if &t_name = T_15_02_04_19_01 or &t_name = T_15_02_04_19_02 %then %do; 

 if txt in ('Mean (SD)' '95% CI of Mean' ) then delete; 

 %end; 

 %else %do; 

 if txt in ('95% CI of Mean') then delete; 

 %end; 

 if avisit = 'Day 5' and atpt = '08:00 PM - 09:30 PM' then do; 

  atptn = 9; 

  avisitn = 105.9; 

 end; 

run; 

 

proc sort data=aval_f; 

 by  paramn param avalu apuper apuperc avisitn avisit atptn atpt txtn txt; 

run; 

 

data final; 

 length period $200.; 

 set aval_f; 

 by paramn param avalu apuper apuperc avisitn avisit atptn atpt txtn txt; 



 if apuper = 1 then do; 

  period = "Period 1"; 

  THS = &N1THS; 

  mcc = &N1mcc; 

  sa = &N1saa; 

 end; 

 else if apuper = 2 then do; 

  period = "Period 2"; 

  ths = &n2ths; 

  mcc = &n2mcc; 

  sa = &n2saa; 

 end; 

 else if apuper = 3 then do; 

  period = "Period 3"; 

  ths = &n3ths; 

  mcc = &n3mcc; 

  sa = &n3saa; 

 end; 

 else if apuper = 4 then do; 

  period = "Period 4"; 

  ths = &n4ths; 

  mcc = &n4mcc; 

  sa = &n4saa; 

 end; 

 



 if txtn in (3,4) and txt ne 'BLOQ, n(%)' then do; 

  if trt_3 ne " " then num_3 = input(trt_3, best.); 

  if trt_4 ne " " then num_4 = input(trt_4, best.); 

  if trt_5 ne " " then num_5 = input(trt_5, best.); 

  if num_3 ne . then pp_3 = strip(put(round((num_3/sa)* 100, 0.1), 15.1)); 

  if num_4 ne . then pp_4 = strip(put(round((num_4/ths)*100, 0.1), 15.1)); 

  if num_5 ne . then pp_5 = strip(put(round((num_5/mcc)*100, 0.1), 15.1)); 

  if trt_3 ne '' then trt_3 = strip(trt_3)||" ("||strip(pp_3)||")"; 

  else trt_3 = '0'; 

  if trt_4 ne '' then trt_4 = strip(trt_4)||" ("||strip(pp_4)||")"; 

  else trt_4 = '0'; 

  if trt_5 ne '' then trt_5 = strip(trt_5)||" ("||strip(pp_5)||")"; 

  else trt_5 = '0'; 

 end; 

 

 drop num_3 num_4 num_5 pp_3 pp_4 pp_5; 

run; 

 

proc sql; 

create table page as 

select distinct apuper, apuperc, paramn, avisitn,atpt 

from final 

order by paramn, apuper, avisitn,atpt; 

quit; 

 



data page; 

 set page; 

 page = _n_; 

run; 

 

proc sql; 

 create table final_page as 

 select distinct a.*, b.page 

 from final as a 

 left join page as b 

 on a.paramn = b.paramn and a.avisitn = b.avisitn and a.apuper = b.apuper and a.atpt=b.atpt 

 order by page, paramn,avalu,apuper,avisitn, atptn,txtn; 

quit; 

 

data final_page; 

 length avisit $200; 

set final_page(rename=(avisit=avisit1)) end=last; 

by page paramn avalu apuper avisitn atptn txtn; 

if last then call symputx("page", page); 

if atpt ne '' then avisit = strip(avisit1)||', '||strip(atpt); 

/*if avisitn in (105 106) and atpt eq '' then do;*/ 

if avisitn = 105.9 or (avisitn = 106 and atpt = '08:00 AM - 09:30 AM') then do; 

 if txt = 'n' then do; 

 if trt_4 = '' then trt_4 = '0'; 

 if trt_5 = '' then trt_5 = '0'; 



 end; 

 else call missing(trt_4,trt_5); 

end; 

else if (avisitn in (105 106) and atpt ne '') or avisitn in (105.1 105.2 105.3 105.4 105.5 105.6 105.7 105.8) 

then do; 

 if txt = 'n' then do; 

 if trt_3 = '' then trt_3 = '0'; 

 end; 

 else call missing(trt_3); 

end; 

 

if txtn = 2 and trt_3 = '0' and trt_4 = '0' and trt_5 = '0' then delete; 

if txtn = 2 and trt_3 = '0' and trt_4 = '' and trt_5 = '' then delete; 

 

 

drop avalu _name_; 

run; 

 

data tflds.&tflno.; 

 set final_page; 

run; 

%if &tflno = T_15_02_04_19_01 or &tflno = T_15_02_04_19_02 %then %do; 

***output dataset for figure; 

proc format; 

value trt 

4='THSm2.2' 



5='mCC' 

3='SA' 

; 

run; 

data tflds.&tflno._f; 

 set aval_log1(drop=mean); 

 %do i=0 %to &last.; 

  if declen=%eval(&i) then do; 

   %let fmt= %sysevalf(12 + (&i +1 )*0.1); 

   %if &i = 0 %then %let fmt1 = 12.; 

   %else  

   %let fmt1= %sysevalf(12 + (&i)*0.1); 

   %let fmt2= %sysevalf(12 + (&i +2 )*0.1); 

  mean= round(mean1,1/10**(declen+1)); 

 end; 

 trtp = put(trtpn,trt.); 

 %end; 

 rename lclmx=lclm uclmx=uclm; 

 keep param paramn avisitn avisit  lclmx uclmx mean trtp trtpn apuper apuperc atpt  atptn; 

run; 

  

proc sort data=tflds.&tflno._f;by paramn apuper avisitn atptn;run; 

%end; 

***output figure dataset end; 

 



 

options number nodate orientation=landscape /* papersize=&P_PGSIZE*/ missing=' ';   

ods escapechar='$'; 

%let linetop = \brdrt\brdrs\brdrw30;  * needs to be 1.5pt so calculated in twips (1/20 pt) ; 

%let linebot = \brdrb\brdrs\brdrw30; 

 

ods path stdlib.t106343 (read) ; 

ods results off; 

ods rtf toc_data/* contents*/ file="/cvn/projects/prj/data/000000106343/TFL/dev/Tables/&tflno..rtf" 

style=t106343 startpage=yes headery=1440 footery=1440 ; 

ods noproctitle; 

%do i=1 %to &page; 

 

 

title ; 

footnote; 

%let wd=0; 

ods proclabel = ' '; 

 

data comp; 

     set final_page end=eof;   

  where page=&i; 

    _firtitl="&t_title";   

  call symput('_FSRTITL', trim(left(_firtitl))); 

  if eof then do; 

     call symput('_FSRTITL', trim(left(_firtitl))); 



/*  call symput('_blankn', compress(put(len,best.)));*/ 

  call symput('period', strip(apuperc)); 

  call symput('param', strip(param)); 

  call symput('N3', strip(put(sa, best.))); 

  call symput('N4', strip(put(ths, best.))); 

  call symput('N5', strip(put(mcc, best.))); 

  end; 

    drop _firtitl ; 

run; 

 

ods listing close; 

 

 

* most set up in template others below; 

* title arial 12pt bold with 12pt paragraph space below; 

* all headers to be arial 11pt bold; 

* data arial 10pt; 

* headers to be central, text values left aligned and numeric centered around decimal point; 

/* Update with your variables as needed */ 

proc report data = comp headline headskip nowd spacing=1 split = '$' %if &i=1 %then %do; contents=' ' 

%end; %else %do; contents='' %end;;; 

 column  page paramn apuper avisitn avisit txtn txt trt_4 trt_5 trt_3; 

  define paramn        / order order = internal noprint; 

  define page        / order order = internal noprint; 

  define avisitn    / order order = internal noprint; 

  define apuper    / order order = internal noprint; 



 

     define txtn   / order order = internal noprint; 

  define avisit           /"Timepoint" order order=internal style={just=left cellwidth=1.5cm} 

style(header)={just=left} ; 

  define txt          /"Statistic" display style={just=left cellwidth=1.3cm} 

style(header)={just=left} ;  

  define trt_4  /"THSm2.2$(N=&N4)" display style={JUST=c cellwidth=1.3cm} 

style(header)={just=center} ;      

  define trt_5           /"mCC$(N=&N5)" display style={just=c cellwidth=1.3cm} 

style(header)={just=center} ; 

  define trt_3           /"SA$(N=&N3)" display style={just=c cellwidth=1.3cm} 

style(header)={just=center}; 

 

 compute after avisitn; 

  line " "; 

 endcomp; 

 

  compute before _page_ / style={just=left protectspecialchars=off};  

     line "\b\fs24\sa24&_FSRTITL." ;  * \b = bold, \fs24 is font size 12pt, \sa24 is space after 12pt; 

  line " "; 

  LINE "Product Use Time Period: &period"; 

  line "&linebot"; 

  endcomp; 

 

  compute before page / style={just=left protectspecialchars=off fontsize=10pt}; 

   line " "; 

  line "&param"; 



   line " "; 

  endcomp; 

 

  compute after _page_/ style={just=left protectspecialchars=off pretext="&linetop."};  

   line 'Note: mCC = Menthol Conventional cigarettes; SA = Smoking abstinence; THSm2.2 

= Tobacco Heating System 2.2 Menthol.'; 

  line "Note: 'Missing' percentages are based on the number of subjects indicated in the 

column header (N), while 'BLOQ' percentages are based on the number of subjects being summarized 

(n)."; 

  line " "; 

  line "&append";  

  line "Study ID: ZRHM-REXA-08-US"  "       "  "Program: &pgname" "       " "Status: &status"  

"                                            " "&sysdate" "                                           " "(Page &i of &page)"; 

  endcomp; 

run; 

%end; 

ods rtf close; 

ods results on; 

ods path sashelp.tmplmst (read); 

 

 

proc datasets kill lib=work memtype=data;run; 

 

%mend; 

 

 

***For table 15.2.4.19.1; 



%t_desc_nc( t_name=T_15_02_04_19_01, 

   dsin=adpc, 

   pop=pp, 

   conf1=%str(if (anl01fl = 'Y' and ((pprot1fl = 'Y' and avisitn in (101 102 103 104 

105 106)) or (pprot2fl = 'Y' and avisitn =130) 

      or (pprot3fl = 'Y' and avisitn = 160) or (pprot4fl = 'Y' and 

avisitn = 190 and atptn ne 10)));), 

   by_var1=paramn paramcd param avisitn avisit, 

   append=%str(Appendix 15.3.3.3), 

   t_title=%str(Table 15.2.4.19.1 Descriptive Statistics of Plasma Nicotine and 

Cotinine Concentrations (ng/mL) - PP Set) 

 

  ); 

 

***For table 15.2.4.19.2; 

%t_desc_nc( t_name=T_15_02_04_19_02, 

   dsin=adpc, 

   pop=fas, 

   conf1=%str(if (anl01fl = 'Y' and ((fasfl = 'Y' and avisitn in (101 102 103 104 105 

106)) or (fasfl = 'Y' and avisitn =130) 

      or (fasfl = 'Y' and avisitn = 160) or (fasfl = 'Y' and avisitn 

= 190 and atptn ne 10)));), 

   by_var1=paramn paramcd param avisitn avisit, 

   append=%str(Appendix 15.3.3.3), 

   t_title=%str(Table 15.2.4.19.2 Descriptive Statistics of Plasma Nicotine and 

Cotinine Concentrations (ng/mL) - FAS) 

 

  ); 



 

proc printto ;                                                                                                               

run;  

ods listing; 

options quotelenmax; 

 

 

 ***check the log, if there is any findings, please make sure to resolve; 

%m_chklog(TFL_part=dev,pgm_type=tables,pgm_name=t_desc_nc,serv=dev,covstudyid=000000106343

); 

 

 


